

A Complex Example

Introduction

The following material traces the development of the extraction optics for our group's high sample temperature mass spectrometer. This serves as an example of how a complex design project can be approached as well as a demonstration of the use and value of geometry files. The resulting lens assembly is composed of a collection of complex 3D parts. Geometry files were used to aid in the design, evaluate the design via simulations, and to create the actual parts drawings that were used to build the actual lens assembly.

In summary, the design path began as a simplified freeform activity to optimize the tradeoffs between the requirements discussed below. The cylindrical nature of the problem permitted the use of Modify created 2D cylindrical potential arrays for quick evaluations and design evolution. After the effort converged on a suitable approach, 3D geometry files were created to evaluate and finalize the design.

Warning!

The geometry files in this example require the use of SIMION 7.0 and create very large potential arrays (up to 40,000,000 points). Your machine should have 500 MB of RAM if you expect view these geometry file generated arrays. Otherwise, you will be forced to run at glacial speeds in virtual memory modes.

Background

The design objectives of the high sample temperature instrument placed significant demands on the ion optics. It was important that the ion extraction optics be capable of supporting electron impact ionization (EI), surface ionization (SI), and secondary ionization mass spectrometry (SIMS), in




Figure 1 Cutaway view of extraction optics

combination or in interleaved modes, and that the ions created by these processes be injected into the quadrupole mass analyzer with roughly equivalent efficiencies.

Each ionization method places operational requirements on the extraction optics that could potentially conflict with the requirements of another method. The challenge was to design extraction optics that satisfied these and other operational requirements without significantly compromising performance.

Ionization Method Issues

For best performance, the electron impact ionization (EI) source needed to pass a dense cloud of electrons very near the front surface of the sample so that the ionized neutrals would have virtually the same focusing characteristics as ions emitted from the sample's surface (SI and SIMS). Moreover, the electron energies should be adjustable between 25-200 eV without significantly moving the region of ionization. The design also needed to provide effective electron sinks to prevent stray electrons from roaming about the optics. There was also a requirement to quickly switch the electron beam on and off via a bias voltage change to facilitate interleaved sampling objectives. Finally, the EI's gun design needed to properly focus the electron beam independent of operator adjustments.

Surface ionization (SI) involves heating the sample through a range of temperatures where some of the molecules volatilized from the sample's surface are ionized. The initial kinetic energies of these thermal ions are normally quite low (< 1.0 eV). Provision must be made within the extraction optics housing for the volume needed for sample heating and temperature measurement devices on the sample probe.

Secondary ion mass spectrometry (SIMS) involves hitting the sample with a high energy (3-10 keV) primary ion beam (in our case Perrhenate – ReO_4^-). The primary beam knocks off both neutrals and secondary ions. The secondary ions have initial energies that range from fractions of an eV to around 20 eV. Holes must be provided in the extraction optics to allow the entry and exit of the primary beam. These holes must not adversely impact the extraction fields or focusing within the extraction optics column. An additional design requirement was to provide a clear primary beam path through the




Figure 2 Extraction of positive ions

extraction optics. This allows the use of a downstream multi-channel plate (MCP) to assist in primary beam aiming via the shadow cast in its image by the sample. Beam fluence measurements are made by pulling back the sample and rotating a small faraday cup into the primary beam's path in front of the MCP. Additional beam holes were also needed for an Argon sputter gun and for visual inspection of the sample. The Argon sputter gun serves to selectively clean or erode the sample's surface.

Design Discussion

These and many other constraints were applied to obtain the final design of the extraction optics shown in Figure 1 above. The basic design makes use of a stacked set of cylindrical electrodes. The stack is held together by rods and separated and positioned by ceramic spacers. The long narrow gaps between electrodes act to prevent any ceramic spacer potential from perturbing the interior field. The stack design has proven advantageous because it can also allow subsequent refinement by changing the design of one or more electrodes in the stack. An outer cylindrical ground shield (not shown) is used to isolate the vacuum chamber from the fields from the electrodes in the stack.

Figure 2 shows a SIMION simulation cross section view with primary beam, electron beam and ion trajectories when positive ions are being extracted. There are four primary beam holes through the extraction optics to support the SIMS beam, Argon sputter beam, and a viewing port for the sample. Notice that the ions of the primary beam that do not hit the sample pass through the assembly to the region of the MCP and Faraday cup described above.

The electron beam is produced by a group of four filaments located along the radial perimeter of the lens assembly. These filaments if used in concert act to create a cylindrically converging electron beam that attains maximum electron density just in front of the sample. The cylindrically converging electron beam design was adopted to improve the probability of ionizing sputtered neutrals. After the electron beam crosses over the sample, it traverses the assembly in a radial direction, passes through a grounded grid and is then trapped by the electron collection electrode.




Figure 3 Potential energy surface view of EI gun

In order to make the EI gun as effective (and user resistant) as possible, a focusing and beam deflection design was adopted that uses electrode geometry (e.g. shapes and positions) rather than tuned electrode

potentials to focus and deflect the electron beam (Figure 3). The advantage of this approach is that the electron beam's trajectory is virtually independent of acceleration voltage (electron kinetic energy). Two sets of collimator slits are employed to further insure consistent electron beam shape. The first collimator is located at the draw out electrode exit and the second is located at the ground potential exit of the gun. A useful feature of this approach is that electrons that impact the exit slit will normally be attracted back to the draw out electrode because it is at a more positive potential. Notice that the leakage ring is positioned to act as an electron shield for any stray electrons that might otherwise hit the sample. Additionally, the design allows the electron gun to be quickly shut off by biasing the filaments a few volts positive relative to the surrounding outer shield electrode.




Figure 4 Potential energy view of ion paths into quadrupole

The potential energy surface view in Figure 4 shows the trajectories of ions emitted from the sample's surface as well as those created by electron bombardment. Notice the low gradient extraction field (flat surface) in the region around the sample. This low gradient field minimizes primary beam deflection and helps to minimize movement of the electron beam when its kinetic energy is changed. Moreover, a low gradient field is highly desirable when a leakage ring is used for self-charge stabilizing the sample's potential¹.

The acceleration and focusing section of the extraction optics is designed to roughly inverse image starting points of the ions at the entrance to the quadrupole. Although one would ideally want all ions to enter the quadrupole almost on axis and with little or no radial kinetic energy, the wide initial ion kinetic energy spreads from SIMS and the relatively broad ion source region from EI led to this focusing compromise.

The extraction optics permit the various ionization modes to operate separately or concurrently. For example, surface ionization (SI) generated ions can be measured by heating the sample with the electron gun and primary beam turned off. Likewise electron impact ionization (EI) of volatilized neutrals (and residual neutrals in the vacuum) can be measured by heating the sample with the EI gun turned on and the leakage ring's potential set to trap the SI ions at the source. SIMS and SI are measured together by heating the sample with the primary beam turned on. EI of sputtered and

thermally volatilized neutrals is accomplished by heating the sample with the EI and SIMS beams on with the leakage ring's potential set to trap SI and SIMS ions at their source. Note the addition of an optional grid across the front of the leakage ring can serve to enhance the rejection of surface emitted ions.

The design of the extraction optics and the ability to quickly switch both the primary beam and the electron beam on and off allows the data system to interleave between these methods as the mass range is scanned by the quadrupole. This is very useful because one measurement can obtain SI, the next SI and SIMS, and the data system can then subtract the two to isolate the estimate of the SIMS spectra. The capabilities of the extraction optics, heated source, the various ionization beams and the method interleaved data system combine to create a versatile and powerful instrument.

Design Path Used

The design path began as a simplified freeform activity and evolved gradually toward the formalized final design and drawings. In the early stages, the only constraints were the general size limitations imposed by the range of dimensions of workable spherical vacuum housings (assuming that the sample would be at the center of the sphere) and the quadrupole mounting requirements.

The highly cylindrical nature of the problem permitted designing with 2D cylindrical potential arrays during most of the design stages. Thus design concepts could be quickly changed via Modify and evaluated. A whole series of design approaches were evaluated before the strategy adopted for the actual lens design was adopted. This approach helped to optimize the tradeoffs between the potentially conflicting ion creation processes described above.

The design developed from the inside out. The inner shapes of the electrodes were determined first, and only in the later stages of design were the details of their external shapes determined. Electrode inner shapes were generally limited to bores and radial faces that can be easily made on a lathe and simulated accurately by SIMION. We have found that this constraint imposes no great limitation on the design and facilitates fabrication.




Figure 5 An early simple 2D extraction lens design

The idea of the cylindrically converging electron beam was thought of early on and was initially simulated with a wire ring in a semi-torrid for ease of simulation (Figure 5 -- not that one could ever

hope to build it). A simple einzel-like focusing arrangement was adopted to extract the ions from the ion formation region on and in front of the sample and roughly focus them as they entered the quadrupole region. This focusing path was intentionally long to minimize beam convergence angles to minimize quadrupole ion insertion losses. Notice the strong focusing at both ends of the extracted ion trajectories. This was not very desirable and was rectified as the design process progressed.




Figure 6 Early example of geometry focused electron gun

The primary focus in the design early stages was to develop a user resistant, cylindrically converging electron gun that could actually be built. Figure 6 serves as an early example of a geometry design that incorporates the electron gun principles used in the final design. However, the rest of the extraction optics looks really grim. The advantage of using 2D and Modify is that designs are easy to make, test, and change. You can learn a lot in a short time.




Figure 7 Design progression with 2D models

As these designs gradually progressed (Figure 7) the extraction optics began to look more like the final design. It was at this point that the design process was converted over to a geometry file based effort. The decision to switch to geometry files was necessitated by the need to model the effects of 3D features like beam holes and filament post effects as well as finalize the electrode layering and external design details.

Conversion to Geometry Files

To facilitate conversion into geometry files a couple of basic ground rules were adopted. First, each electrode was to be defined in physical units at the origin and then shifted to its proper location within the volume. Second, all electrode dimensions were to snap (index) to 0.010" spacing. The 0.010" spacing was chosen in view of SIMION 7.0's potential array memory limits (50,000,000 points). The desired four primary beam holes gave the design a 90 degree rotational symmetry. This meant that a 3D planar array model could be used with y and z mirroring (1/4th of the total points). However, given the size of the extraction lens, the 0.010" grid spacing was the smallest practical spacing (requiring 40,000,000 points or 400 MB of RAM for the array).




Figure 8 External details – Exit Ground Electrode.gem

The first step in the conversion process was to make a 2D array that internally conformed to the 0.010" spacing as a starting point. This allowed the electrode geometry definitions to be initially constrained by these internal dimensions. The remaining issues revolved around how to layer successive electrodes and mount them together. Because of the 90° symmetry of the design it was decided to use a four mounting rod approach. These four rods were rotated 45° from the beam holes (midway between them – Figure 8). The choice was made to use hollow ceramic spacers to separate and align the lens elements. The rods would run through the ceramic spacers to create an insulated packed sandwich design. Since insulators can often charge (or accidentally be precharged) to high potentials, it was decided to intentionally create long gap paths between electrodes to effectively eliminate any adverse impact of external insulator charging on the internal fields of the extraction lens.

Another external design issue involved the need to route wires to the lens electrodes. Early on it was decided that the extraction optics should have an external ground shield to prevent unwanted incoming

primary beam deflections due to changes in electrode potentials. Moreover, it was highly desirable that the extraction optics and quadrupole be one assembly that could be setup on the bench and lowered as a single assembly into the instrument. This dictated that all wires needed to run under the ground shield and exit at the quadrupole end of the extraction optics. The solution was to provide parallel slots in the outer electrode flanges between the beam holes and assembly rods (8 slots in total – centered every 45° between an adjacent beam hole and rod – Figure 8). The real problem occurred around the electron gun. Unfortunately it extended out almost to the shield. This precluded the slot approach. The solution was to use the four insulated threaded rods that held that end of the lens together as electrical conductors to pass potentials through the electron gun to the electrodes on the left (sample side) of the electron gun filament housing.

Now that these conceptual design issues were resolved the detailed design of each electrode could begin. The electrode strategy was to define the geometry of one electrode at a time starting from right (quad. exit electrode) and work left. Details of the location and details of the interface gaps between successive electrodes resolved during this definition process.

```

*****
: ***** right quad mounting ground electrode *****
: *****
:
locate(0,0,0,1,-90) ; swing back 90 deg cw in azimuth and
{
  locate(0,0,0,1,90) ; swing 90 deg ccw in azimuth
  {
    rotate_fill(180){ ; create 1/2 volume of revolution
      within(polyline( 0.000, 1.080, ; defines cross section of
        0.000, 1.800, ; electrode
        0.200, 1.800,
        0.200, 1.400,
        0.410, 1.400,
        0.410, 0.550,
        0.160, 0.550,
        0.160, 1.080))
    }
  }
}
n(0) ; erase with zero volt non-electrode points
{
  fill{ ; community erasing fill
    locate(0,0,0,1,0,-45) ; swing 45 deg cw in elevation
    {
      locate(0,1.65) ; shift up in y to radius to use
      {
        ; define set of mounting holes
        within_inside(cylinder(0,0,0.001,100,0.051)) ; left insulator mount
        within_inside(cylinder(0,0,0.001,0.65,0.202)) ; mounting hole
        within_inside(cylinder(0,0,-150,100,0.051)) ; right insulator mount
      }
    }
    locate(0,0,0,1,0,-135) ; swing 135 deg cw in elevation
    {
      locate(0,1.65) ; shift up in y to radius to use
      {
        ; define set of mounting holes
        within_inside(cylinder(0,0,0.001,100,0.051)) ; left insulator mount
        within_inside(cylinder(0,0,0.001,0.65,0.202)) ; mounting hole
        within_inside(cylinder(0,0,-150,100,0.051)) ; right insulator mount
      }
    }
    locate(0,0,0,1,0,0) ; bore hole in top of shoulder
    {
      locate(0,1.40,-31,1,0,0,-90) ; position to bore down into shoulder
      {
        within_inside(cylinder(0,0,0.001,0.40,0.40)) ; Quad attachment holes
      }
    }
    locate(0,0,0,1,0,-90) ; swing 90 deg cw to bore hole in right
    {
      ; edge of shoulder
      locate(0,1.40,-31,1,0,0,-90) ; position to bore down into shoulder
      {
        within_inside(cylinder(0,0,0.001,0.40,0.40)) ; Quad attachment holes
      }
    }
    locate(0,0,0,1,0,-22.5) ; swing cw 22.5 degs from top
    {
      within_inside(box3d(-3,1.5,-201,3,2,0.001)) ; wiring slot
    }
    locate(0,0,0,1,0,-67.5) ; swing cw 67.5 degs from top
    {
      within_inside(box3d(-3,1.5,-201,3,2,0.001)) ; wiring slot
    }
    locate(0,0,0,1,0,-112.5) ; swing cw 112.5 degs from top
    {
      within_inside(box3d(-3,1.5,-201,3,2,0.001)) ; wiring slot
    }
  }
}
}

```


Electrodes were defined in physical units (inches) at the origin and *then* positioned in the physical volume. Because of the rotational symmetry of the electrodes a rotate fill using a polyline outline of the cross section of the electrode created the initial machining blank of the electrode. This blank was then rotated 90° in azimuth so that slots and mounting holes and shoulders could be drilled using zero potential non-electrode points (e.g. n(0){}). The file image above is that of the include file used to create the Exit Ground Electrode in Figure 8 (z_include_exitground.gem).

You will notice that the rotate fill is set for 180 degrees. While the actual array volume that the definition fills is only 90 degrees, there was a desire to rotate the extraction optics about the x axis to selectively align either the mounting holes or primary beam holes in the y direction for selective use and display in both 2D and 3D arrays. Thus enough extra elements had to be defined so that the image could be rotated 45 degrees without getting into any undefined volume. This insured that both 2D and 3D images of the extraction optics would remain fully defined in this 45 degree rotation region.

Notice that the mounting rod hole and ceramic insulator shoulders are created at the origin as three cylinder calls. This facilitates the definition of these features and the cylinder command limits the extent of the holes (we don't want infinite holes like a circle would make). Once defined, a locate shifts the holes up in y to their location circle ($r = 1.65''$). The next outer locate then rotates them in elevation along the mounting circle to the proper angles (-45 and -135 degrees). Notice that there are two groups of holes defined. One at -45 degrees for a non-rotated view, and the other at -135 degrees for providing the needed holes for a 3D mounting hole aligned orientation. The second set of holes were made simply by making a second copy of the first set of calls and changing the angle (pretty easy).

The wiring slots were defined by a box3d command already located in positive y at the required mounting circle. A locate command is then used to swing each slot to its assigned location. As noted above each mounting slot definition is simply a copy of the original slot definition with the swing angle definition changed (elevation angle). Note that one additional slot is define to permit the 45 degree orientation change desired.

Two quad attachment holes are also defined by use of the swinging drill technique of two locates. Study these two calls to see if you can understand how they work.

The actual design process consisted of creating a giant gem file containing all the lens elements in it (e.g. Assembly in Single File.gem). If you examine the file with an editor you will see the following header:

```
;This gem file creates the ion extraction optics for our group's
;high temperature instrument. The gem file creates a 2D potential
;array by default size = 724,563 points.

;If you have lots of RAM (500 megs) in your machine you can try the 3D version
;1. Comment out the two 2d array definition lines.
;2. Uncomment the 3D's pa_define line and the beam hole aligned locate line

;2d array definition
  pa_define(1953,371,1,c,y)
; scaling is 0.005"/grid unit or 0.127 mm/grid unit
  locate(1952,0,0,200,,,0) ; beam hole aligned
;   locate(1952,0,0,200,,, -11.25) ; middle of filament
;   locate(1952,0,0,200,,, -45) ; mounting hole aligned

; 3d array definition
;   pa_define(977,186,186,p,yz)
; scaling is 0.010"/grid unit or 0.254 mm/grid unit
;   locate(976,0,0,100,0,0,0) ; beam hole aligned
;   locate(976,0,0,100,0,0,22.5) ; filament end aligned
;   locate(976,0,0,100,0,0, -45) ; mounting hole aligned

;note:   above locates position origin on right edge of array
;        lens elements are built from right to left with each
;        successive element shifted more to the left than the
;        preceeding
```

This header contains both 2D array and 3D array pa_defines commands along with a selected number of locate commands that provide optional orientation alignments (e.g. beam hole alignment). Thus one gem file can be used to create a collection of arrays and alignments by simply commenting and uncommenting the appropriate commands. This is very powerful and useful during the development process. The 3D arrays initially served more as an interference and physical appearance monitoring check. The 2D arrays were used in simulations to check the impact of the many small inside dimension changes that were required as the design progressed toward completion.

Once satisfactory 2D simulations were obtained, 3D simulations (long refine times and big disk usage) were done to evaluate the effects of the filament designs and the primary beam holes. These simulations verified that the initial 2D simulations were quite accurate (as expected), and that the extraction optics would probably work according to simulations (subsequently verified with the as built system).

The primary task remaining was to generate the parts drawings and have the extraction optics made. It was decided to use SIMION to create the drawings, because the model shop was reasonably flexible and SIMION 7.0 could create dimensions right off the drawings. The first step in the process was to separate each electrode into its own include file (e.g. z_include_exitground.gem). A single electrode generation calling file was then created to project the include file into an appropriately sized array (e.g. Exit Ground Electrode.gem). An assembly include base gem file was also created to verify that the include file electrodes were defined and fit together properly (e.g. Assembly With Includes.gem). The use of include files also opens up the opportunity for exploded views (Figure 9 – I digress):




Figure 9 Exploded Electron Gun.gem

The include files were also very useful in the design of a filament jig holder (Filament Jig.gem) that was required to precisely hold the filaments for spot welding on the filament posts mounted in the left filament holder electrode. The filament and post include file (z_include_filament.gem) and the left filament holder electrode include file (z_include_lft_fil_hold.gem) were expanded to full 180 degree

definition files so that two calls to each (rotated 180 degrees) could create the full 360 degree definitions of them required to generate the in use example (Filament Jig in Use.gem) shown in Figure 10 below:




Figure 10 Filament Jig in Use.gem

Once the gem files were created for the individual electrodes and jigs, SIMION 7.0 was used to create the parts drawings. The gem file creates the 3D array for the electrode or part, then its array instance in View was scaled to 0.254 mm/gu or 0.010 inches per grid unit. Cursor coordinates were set to instance 1 in inches and relative or absolute coordinates were used as needed.


To make a parts drawing the electrode was oriented, sized, and positioned as desired in View. The print function was then used to annotate the image as required. The dimension option within 7.0 was used to create all dimensions. Note that the length of the dimensions are based on pixel resolutions. However, the screen resolution used was 1600x1200 and most dimensions came out exactly or close. Since the dimension number value is a label, its value and format can be easily changed to give the correct and desired value. Moreover, the label can be freely moved to the most appropriate spot for its viewing.

In general, at least one dimensioned end and side view were created for each electrode as well as one or two isometric views to help clarify to the machinist what the part would actually look like. The following pages contain the images of the drawings for the exit ground electrode.


ⁱ D.A. Dahl, A.D. Appelhans, Int. J. Mass Spectrom. 178 (1998) 187-204.

Ground Exit Plate

304 Stainless Steel



(Cutaway View)



Ground Exit Plate

304 Stainless Steel

* means critical Dimension

Angles are Critical
Between Holes

90 degree *
Hole Spacings

